

可视化程序设计

课程 实验 指导 书

撰写人：***

日期：2014年3月12日

实验一 一个简单 VB 应用程序的创建和命令按钮、标签及文本框的应用

日期：2 月 23 日

一、实验目的与要求：

1. 掌握 VB 应用程序的创建步骤；
2. 了解 VB 应用程序的组成；
3. 掌握命令按钮、标签和文本框三个基本控件的属性、事件和方法的使用。

二、实验内容：

设计一程序，当程序运行后，在窗体中显示"你好!请输入姓名"，焦点定在其下的文本框中，如图 1-1 所示界面。



图 1-1

当用户输入姓名并单击"确定"按钮后，在窗体的正中间用黑体、三号、红色显示"XXX 同学,你好!祝你学好 VB 程序设计"，同时窗体上出现两个命令按钮"继续"和"结束"，其中"XXX"是用户输入的姓名。

例如：当用户输入"王五"，单击"确定"按钮后，出现另如图 1-2 所示界面，如果单击"继续"按钮，则又回到初始运行状态；单击"结束"按钮即结束程序运行。




图 1-2

三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

- (1) 设计应用程序的界面：在窗体上创建三个命令按钮 Command1、Command2 和 Command3、两个标签 Label1 和 Label2、一个文本框 Text1。

(2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	实验 1-1
Command1	Caption	确定
Command2	Caption	继续
Command3	Caption	结束
Label1	Caption	你好! 请输入你的姓名
Label2	Caption	空串
Text1	Text	空串

(3) 编写事件代码:

```
Private Sub Form_Load()  
    Command2.Visible = False  
    Command3.Visible = False  
    Label2.Visible = False  
End Sub  
Private Sub Command1_Click()  
    Command1.Visible = False  
    Text1.Visible = False  
    Label1.Visible = False  
    Label2.Visible = True  
    Label2.AutoSize = True  
    Label2.FontName = "黑体"  
    Label2.ForeColor = vbRed  
    Label2.Caption = Text1.Text & "同学, 你好! 祝你学好 VB 程序设计"  
    Command2.Visible = True  
    Command3.Visible = True  
End Sub  
Private Sub Command2_Click()  
    Command2.Visible = False  
    Command3.Visible = False  
    Label1.Visible = True  
    Label2.Visible = False  
    Command1.Visible = True  
    Text1.Visible = True  
End Sub  
Private Sub Command3_Click()  
    End  
End Sub
```


(4) 程序的运行及保存。

五、实验结果：

在文本框中输入：王五



单击“确定”按钮后，则显示如下：



六、实验小结：（要能回答如下问题）

1. 创建 VB 应用程序的步骤是什么？
2. 窗体常用属性、事件和方法有哪些？
3. 命令按钮、标签和文本框控件的常用属性、事件和方法有哪些？

一、实验目的与要求：

1. 了解 VB 顺序结构程序设计的特点；
2. 掌握常用 VB 运算符及表达式和相关内部函数的使用。

二、实验内容：

- 1、编写程序，界面如图 2-1 所示。使得单击一次按钮可以产生一个[60, 90]之间的随机数并显示在标签 Label1 上，再求出该数的正弦值，将结果写在标签 Label2 上。




图 2-1 实验 2-2 程序运行界面

- 2、编写一个华氏温度 F 与摄氏温度 C 之间转换的应用程序，界面如图 2-2 所示。其中

F 与 C 之间的关系为：
$$F = \frac{9}{5}C + 32$$



图 2-2 实验 2-3 的运行界面图

三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

实验 1:

- (1) 设计应用程序的界面：在窗体上创建一个命令按钮 Command1、两个标签 Label1 和 Label2。
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	随机数的正弦值
Command1	Caption	下一个随机数
Label1	Caption	
Label2	Caption	

- (3) 编写事件代码:

```
Private Sub Command1_Click()  
    Dim x!, y!  
    Randomize  
    x = Rnd * 31 + 60  
    y = Sin(x)  
    Label1.Caption = "随机数 (60-90) : " & Str(x)  
    Label2.Caption = "正弦值          : " & Str(y)  
End Sub
```

- (4) 程序的运行及保存。

实验 2:

- (1) 设计应用程序的界面：在窗体上创建两个命令按钮 Command1 和 Command2、两个标签 Label1 和 Label2、两个文本框 Text1 和 Text2。
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	温度转换
Command1	Caption	>
Command2	Caption	<
Label1	Caption	华氏温度
Label2	Caption	摄氏温度
Text1	Text	空串
Text2	Text	空串

- (3) 编写事件代码:

```
Private Sub Command1_Click()  
    Dim f!, c!  
    f = Val(Text1.Text)  
    c = (f - 32) * 5 / 9  
    Text2.Text = Str(c)  
End Sub
```

```
Private Sub Command2_Click()  
    Dim f!, c!  
    c = Val(Text2.Text)  
    f = 9 / 5 * c + 32  
    Text1.Text = Str(f)  
End Sub
```

(4) 程序的运行及保存。

五、实验结果：

已经显示在窗体上。

六、实验小结：（要能回答如下问题）


1. 怎样产生一个[60, 90]之间的随机数？
2. 怎样求正弦值？
3. 怎样防止两次运行程序的随机数序列相同？

一、实验目的与要求：

1. 了解 VB 选择结构程序设计的特点；
2. 掌握 If 语句的格式及其使用；
3. 掌握 Select Case 语句的格式及其使用。


二、实验内容：

1、设计一个求解一元二次方程 $ax^2+bx+c=0$ 的根的程序,要求考虑实根、虚根情况。界面如下所示:



2、设计"健康称"程序，具体要求如下：

- (1)将两个文本框的文字对齐方式均设置为右对齐，最多接收 3 个字符；
- (2)两个文本框均不接收非数字字符；
- (3)单击"健康状况"按钮后,根据计算公式将相应提示信息通过标签显示。



计算公式为:标准体重=身高-105

体重高于体重的 1.1 倍为偏胖, 提示"偏胖,注意饮食";

体重低于标准体重的 90%为偏瘦,提示"偏瘦,增加营养";

其他为正常,提示"正常,继续保持".

三、实验器材：

四、实验步骤:

实验 1:

- (1) 设计应用程序的界面: 在窗体上创建一个命令按钮 Command1、五个标签 Label1、Label2、Label3、Label4 和 Label5、五个文本框 Text1、Text2、Text3、Text4 和 Text5.
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	一元二次方程
Command1	Caption	求解
Label1	Caption	$x^2 +$
Label2	Caption	$x +$
Label3	Caption	$= 0$
Label4	Caption	x_1
Label5	Caption	x_2
Text1	Text	空串
Text2	Text	空串
Text3	Text	空串
Text4	Text	空串
Text5	Text	空串

- (3) 编写事件代码:

```
Private Sub Command1_Click()
    Dim a!, b!, c!, x1!, x2!, disc!
    a = Val(Trim(Text1.Text))
    b = Val(Trim(Text2.Text))
    c = Val(Trim(Text3.Text))
    disc = b * b - 4 * a * c
    If disc >= 0 Then ' 实根情况
        x1 = (-b + Sqr(disc)) / (2 * a)
        x2 = (-b - Sqr(disc)) / (2 * a)
        Text4.Text = Str$(x1)
        Text5.Text = Str$(x2)
    Else ' 虚根情况
        x1 = -b / (2 * a)
        x2 = Sqr(Abs(disc)) / (2 * a)
        Text4.Text = Str$(x1) & "+" & Str$(x2) & "i"
        Text5.Text = Str$(x1) & "-" & Str$(x2) & "i"
    End If
End Sub
```

- (4) 程序的运行及保存。

实验 2:

- (1) 设计应用程序的界面: 在窗体上创建一个命令按钮 Command1、五个标签 Label1、Label2、Label3、Label4 和 Label5、两个文本框 Text1 和 Text2。
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	健康称
Form1	BorderStyle	3-Fixed Dialog
Command1	Caption	健康状况
Label1	Caption	身高:
Label2	Caption	体重:
Label3	Caption	cm
Label4	Caption	kg
Label5	Caption	空串
Text1	Text	空串
Text2	Text	空串

(3) 编写事件代码:

```
Private Sub Command1_Click()
    Dim bzh As Single
    bzh = Val(Text1.Text) - 105
    If Val(Text2.Text) > bzh * 1.1 Then
        Label5.Caption = "偏胖, 注意饮食"
    ElseIf Val(Text2.Text) < bzh * 0.9 Then
        Label5.Caption = "偏瘦, 增加营养"
    Else
        Label5.Caption = "正常, 继续保持"
    End If
End Sub
```

(4) 程序的运行及保存。

五、实验结果:

已经显示在窗体上。

六、实验小结:(要能回答如下问题)


1. If 语句的格式是什么?
2. Select Case 语句的格式是什么? 将实验 2 改写成 Select Case 语句的格式。
3. 怎样设置窗体的标题栏上不显示最小化和最大化按钮?

一、实验目的与要求：

1. 了解 VB 循环结构程序设计的特点；
2. 掌握 For...Next 语句、Do...Loop 语句和 While...Wend 语句的格式及其使用；
3. 学会使用 InputBox 函数和 MsgBox 函数进行输入和输出。

二、实验内容：

编程实现：在窗体上，打印由数字组成的如下图所示的金子塔图案。



三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

- (1) 设计应用程序的界面：在窗体上创建一个命令按钮 Command1、一个标签 Label1、一个文本框 Text1。
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	金字塔图案
Command1	Caption	确定
Label1	Caption	输入一个整数 n (n 大于 0、小于 10)
Text1	Text	空串

- (3) 编写事件代码：

```
Private Sub Command1_Click()
    Dim i%, j%, n%
    Form1.Cls
```

```

n = Val(Text1.Text)
If n <= 0 Or n >= 10 Then
    MsgBox "输入错误,请重新输入", vbInformation, "提示信息"
    Text1 = ""
    Text1.SetFocus
    Exit Sub
End If
For i = 1 To n                '外循环控制打印行数
    If i <= n / 2 Then
        Print Tab(10 - i);    '每行起始打印位置
        For j = 1 To 2 * i - 1 '内循环控制打印个数
            Print Trim(Str(i)); '打印内容
        Next j
        Print
    Else
        k = n - i + 1
        Print Tab(10 - k);
        For j = 1 To 2 * k - 1
            Print Trim(Str(i));
        Next j
        Print
    End If
Next i
End Sub

Private Sub Text1_KeyPress(KeyAscii As Integer)
    If KeyAscii = 13 Then '按回车键调用"判断"命令按钮
        Call Command1_Click
        Exit Sub
    End If
    Select Case Chr(KeyAscii)
        Case "0" To "9", ",", Chr(8)
        Case Else
            KeyAscii = 0
    End Select
End Sub

```

(4) 程序的运行及保存。

五、实验结果：

已经显示在窗体上。

六、实验小结：（要能回答如下问题）

1. 怎样在窗体上打印或输出表达式的结果？
2. 怎样改写代码用 InputBox 函数和 MsgBox 函数进行输入和输出？
3. 怎样禁止在文本框中只能输入 0 到 9 之间的数字字符？


一、实验目的与要求：

1. 掌握一维数组和二维数组的定义格式和引用；
2. 掌握一维数组和二维数组的基本操作（输入、输出、求最大值、排序等）；
3. 学会使用一维数组或二维数组解决相关问题。

二、实验内容：

编写一个双色球摇号程序要求：

- (1) 随机生成 6 个 1 到 33 之间的正整数和一个 1 到 16 之间的正整数
- (2) 6 个红色球不能相同,且从小到大排好序.



三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

- (1) 设计应用程序的界面：在窗体上创建一个命令按钮 Command1、两个标签 Label1 和 Label2、七个文本框 Text1、Text2、Text3、Text4、Text5、Text6 和 Text7.
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	双色球摇号程序
Command1	Caption	摇号
Label1	Caption	红色球: (1-33)
Label1	ForeColor	vbred
Label2	Caption	蓝色球: (1-16)
Label2	ForeColor	vbblue
Text1	Text	空串
Text2	Text	空串

Text3	Text	空串
Text4	Text	空串
Text5	Text	空串
Text6	Text	空串
Text7	Text	空串

(3) 编写事件代码:

```
Private Sub Command1_Click()
    Dim a(5) As Integer, x As Integer
    Dim i As Integer, j As Integer
    Randomize
        '生成6个不同的1到33之间的随机整数存放于a数组中
    a(0) = Int(Rnd * 33) + 1
    For i = 1 To 5
    L1:   x = Int(Rnd * 33) + 1
        For j = 0 To i - 1
            If x = a(j) Then
                GoTo L1
            Exit For
        End If
    Next j
    a(i) = x
    Next i
    For i = 0 To 4
        For j = i + 1 To 5
            If a(i) > a(j) Then
                t = a(i)
                a(i) = a(j)
                a(j) = t
            End If
        Next j
    Next i
    Text1.Text = a(0)
    Text2.Text = a(1)
    Text3.Text = a(2)
    Text4.Text = a(3)
    Text5.Text = a(4)
    Text6.Text = a(5)
    Text7.Text = Int(Rnd * 16) + 1
        '生成一个1到16之间的随机整数
End Sub
```

(4) 程序的运行及保存。

五、实验结果:

已经显示在窗体上。

六、实验小结:(要能回答如下问题)

1. 怎样随机产生6个完全不同的1到33之间的正整数?
2. 怎样对数组中的6个数进行从小到大排序?(选择排序或冒泡排序)
3. 怎样使用控件数组?将上述实验改写成用控件数组来实现。


一、实验目的与要求：

1. 掌握动态数组的定义格式和引用；
2. 掌握与数组操作相关的几个函数的应用；
3. 学会使用动态数组设计相关应用程序。

二、实验内容：

输入一系列英文单词，按升序排列输出。要求：

- (1) 在文本框 text1 中每输入一个英文单词，按回车键后即把该单词存放到数组中，并输出在排序前的文本框 text2 中；
- (2) 单击“排序”命令按钮，字符数组升序排序，并在排序后的文本框 text3 中输出。
- (3) 程序运行后窗体界面如下所示：



三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

- (1) 设计应用程序的界面：在窗体上创建一个命令按钮 Command1、三个标签 Label1、Label2 和 Label3、三个文本框 Text1、Text2 和 Text3。
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	单词排序
Command1	Caption	排序
Label1	Caption	输入单词：
Label2	Caption	排序前：
Label3	Caption	排序后：

Text1	Text	空串
Text2	Text	空串
Text2	MultiLine	True
Text2	ScrollBars	2-Vertical
Text3	Text	空串
Text3	MultiLine	True
Text3	ScrollBars	2-Vertical

(3) 编写事件代码:

```

Dim a() As String, n%

Private Sub Command1_Click()
    For i = 1 To n - 1
        For j = i + 1 To n
            If a(i) > a(j) Then
                t = a(i)
                a(i) = a(j)
                a(j) = t
            End If
        Next j
    Next i
    For i = 1 To n
        Text3.Text = Text3.Text & a(i) & vbCrLf
    Next i
End Sub

Private Sub Text1_KeyPress(KeyAscii As Integer)
    If KeyAscii = 13 Then
        n = n + 1
        ReDim Preserve a(n)
        '动态存储数组 a 增加 1 个元素, 前 n-1 个元素值保持不变。
        a(n) = Text1.Text
        Text2.Text = Text2.Text + Text1.Text & vbCrLf
        Text1.Text = ""
    End If
End Sub

```

(4) 程序的运行及保存。

五、实验结果:

已经显示在窗体上。

六、实验小结:(要能回答如下问题)

1. 怎样定义动态数组?
2. 定义动态数组时, Preserve 的作用是什么?
3. 怎样对各个字符串进行排序?

一、实验目的与要求：

1. 掌握 Sub 过程和函数过程的定义格式及调用；
2. 了解过程之间的参数传递（传址和传值）；
3. 掌握过程与变量的作用域。

二、实验内容：

自定义一 Sub 过程：实现把一个给定数插入到有序数列中，插入后数列仍然有序。

编程：针对窗体的单击事件中编写程序代码，验证上述过程的正确性。

三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

(1) 设计应用程序的界面：新建一新工程即可，无需添加任何控件。

(2) 编程分析：设 n 个有序数（从小到大）存放在数组 $a(1)$ - $a(n)$ 中，要插入的数为 x 。首先，确定 x 插在数组中的位置 p ，假设要在一个具有 n 个升序排列元素的一维数组中插入一个新的元素 k ，算法如下：

- ① 从第 1 个元素开始逐个与 k 比较，一旦发现第 p 个元素大于 x ，则确定插入的位置为 p ，如果所有元素均小于 x ，则确定插入的位置为 $n+1$ 。
- ② 重新定义数组大小，从第 n 个元素到第 p 个元素逐一向后移动一个位置。
- ③ 将 x 赋值给第 p 个元素，完成插入操作。

(3) 编写自定义过程和窗体的单击事件代码：

(4) 程序的运行及保存。

Option Base 1

```
Private Sub Instert(a(), x As Single)
    Dim p%, n%, i%
    n = UBound(a)
    ReDim Preserve a(n + 1) ' 让数组长度增加 1，以便存放插入的数
    p = 1
    Do While x > a(p) And p <= n ' 确定 x 应插入的位置
        p = p + 1
    Loop
    For i = n To p Step -1
        a(i + 1) = a(i)
    Next i
    a(p) = x
End Sub
```

```
Private Sub Form_Click()
    Dim i As Integer, x()
    Dim y As Single, k As Integer
    x = Array(1, 4, 8, 10, 20, 30, 40, 46, 50, 55, 60, 64)
    For i = LBound(x) To UBound(x) ' 打印输出插入前的数据
        Print x(i);
    Next i
    Print
```

```

y = Val (InputBox("输入要插入的数"))
Call Instert(x, y)      '调用插入子过程实现插入操作
For i = LBound(x) To UBound(x)  '打印输出插入后的数据
    Print x(i);
Next i
End Sub



```

五、实验结果：

原序列为：1,4,8,10,20,30,40,46,50,55,60,64

插入为：7

则输出为：1,4,7,8,10,20,30,40,46,50,55,60,64



六、实验小结：（要能回答如下问题）

1. 若 n 个有序数（从大到小）存放在数组中，如何改写 Insert()过程？
2. 如何改写 Insert()过程，使其既适合数组由小到大排序又适合由大到小排序的情况？
3. 如果调用时，数组 x 无序，则程序运行后出现什么情况？

一、实验目的与要求：

1. 掌握单选按钮、复选框、框架及组合框、列表框等控件的属性、事件和方法的使用；
2. 学会使用各种常用控件设计相关 VB 实用应用程序。

二、实验内容：

设计一个个人资料输入窗口，使用单选按钮选择“性别”，组合框选择“民族”和“职业”，检查框选择“爱好”，当单击“确定”按钮，列表框列出个人资料信息，程序运行界面如下：



三、实验器材：

微机+windows 操作系统+VB 6.0

四、实验步骤：

- (1) 设计应用程序的界面：在窗体上创建三个命令按钮 Command1、Command2 和 Command3、五个标签 Label1、Label2、Label3、Label4 和 Label5、两个文本框 Text1 和 Text2、两个组合框 Combo1 和 Combo2、一个列表框 List1、两个单选按钮 Option1 和 Option2、两个框架 Frame1 和 Frame2、四个检查框 Check1、Check2、Check3 和 Check4。
- (2) 设置对象的属性

对象名称	属性名称	属性值
Form1	Caption	个人资料
Command1	Caption	确定
Command2	Caption	重选
Command2	Caption	上交
Label1	Caption	姓名：
Label2	Caption	民族：
Label3	Caption	个人资料：
Label4	Caption	年龄：

以上内容仅为本文档的试下载部分，为可阅读页数的一半内容。

如要下载或阅读全文，请访问：

<https://d.book118.com/325121011032011304>